

# CafeOBJ Syntax Quick Reference

for Interpreter version 1.4.8 or later

Toshimi Sawada\*  
Kosakusha Inc.

September 18, 2014

## Contents

<b>1</b>	<b>Syntax</b>	<b>1</b>
1.1	CafeOBJ Spec . . . . .	2
1.2	Module Declaration . . . . .	2
1.3	Module Expression . . . . .	3
1.4	View Declaration . . . . .	3
1.5	Evaluation . . . . .	3
1.6	Sugars and Abbreviations . . . . .	3
<b>2</b>	<b>Lexical Considerations</b>	<b>5</b>
2.1	Reserved Word . . . . .	6
2.2	Self-terminating Characters . . . . .	6
2.3	Identifier . . . . .	6
2.4	Operator Symbol . . . . .	6
2.5	Comments and Separators . . . . .	6

## 1 Syntax

We use an extended BNF grammar to define the syntax. The general form of a production is

*nonterminal* ::= *alternative* | *alternative* | ··· | *alternative*

The following extensions are used:

a ···	a list of one or more <i>as</i> .
a, ···	a list of one or more <i>as</i> separated by commas: “a” or “a, a” or “a, a, a”, etc.
{ a }	{ and } are meta-syntactical brackets treating <i>a</i> as one syntactic category.
[ a ]	an optional <i>a</i> : “ ” or “a”.

Nonterminal symbols appear in *italic face*. Terminal symbols appear in the face like this: “terminal”, and may be surrounded by “ and ” for emphasis or to avoid confusion with meta characters used in the extended BNF. We will refer terminal symbols other than self-terminating characters (see section 2.2) as *keywords* in this document.

---

\* sawada@cafeobj.org

## 1.1 CafeOBJ Spec

*spec ::= { module | view | eval } ...*

A **CafeOBJ** spec is a sequence of *module* (module declaration – section 1.2), *view* (view declaration – section 1.4) or *eval* (*reduce* or *execute* term – section 1.5).

## 1.2 Module Declaration

<i>module</i>	$::= \text{module\_type} \text{ module\_name} [\text{parameters}] [\text{principal\_sort}]$	
	$\{"\text{ module\_elt} \dots "\}$	
<i>module_type</i>	$::= \text{module}   \text{module!}   \text{module}^*$	<sup>– 1</sup>
<i>module_name</i>	$::= \text{ident}$	
<i>parameters</i>	$::= ("(\text{ parameter}, \dots ")$	
<i>parameter</i>	$::= [\text{protecting}   \text{extending}   \text{including}] \text{ paramter\_name} :: \text{module\_expr}$	<sup>– 23</sup>
<i>parameter_name</i>	$::= \text{ident}$	
<i>principal_sort</i>	$::= \text{principal\_sort} \text{ sort\_name}$	
<i>module_elt</i>	$::= \text{import}   \text{sort}   \text{operator}   \text{variable}   \text{axiom}   \text{macro}   \text{comment}$	<sup>– 4</sup>
<i>import</i>	$::= \{ \text{protecting}   \text{extending}   \text{including}   \text{using} \} ("(\text{ module\_expr })")$	
<i>sort</i>	$::= \text{visible\_sort}   \text{hidden\_sort}$	
<i>visible_sort</i>	$::= "[" \text{ sort\_decl}, \dots "]"$	
<i>hidden_sort</i>	$::= "*"[" \text{ sort\_decl}, \dots "]*"'$	
<i>sort_decl</i>	$::= \text{sort\_name} \dots [\text{supersorts} \dots ]$	
<i>supersorts</i>	$::= < \text{sort\_name} \dots$	
<i>sort_name</i>	$::= \text{sort\_symbol}[\text{ qualifier}]$	<sup>– 5</sup>
<i>sort_symbol</i>	$::= \text{ident}$	
<i>qualifier</i>	$::= ":" \text{module\_name}$	
<i>operator</i>	$::= \{ \text{op}   \text{bop} \} \text{ operator\_symbol} : [\text{arity}] \rightarrow \text{coarity} [\text{ op\_attrs}]$	<sup>– 6</sup>
<i>arity</i>	$::= \text{sort\_name} \dots$	
<i>coarity</i>	$::= \text{sort\_name}$	
<i>op_attrs</i>	$::= \{"\text{ op\_attr} \dots "\}$	
<i>op_attr</i>	$::= \text{constr}   \text{associative}   \text{commutative}   \text{idempotent}   \{ \text{id:}   \text{idr:} \} ("(\text{ term })")$   $\text{strat:} ("(\text{ natural} \dots ")")$   $\text{prec:} \text{ natural}   \text{l-assoc}   \text{r-assoc}   \text{coherent}   \text{demod}$	<sup>– 7</sup>
<i>variable</i>	$::= \text{var} \text{ var\_name} : \text{sort\_name}   \text{vars} \text{ var\_name} \dots : \text{sort\_name}$	
<i>var_name</i>	$::= \text{ident}$	
<i>axiom</i>	$::= \text{equation}   \text{cequation}   \text{transition}   \text{ctransition}   \text{fol}$	
<i>equation</i>	$::= \{ \text{eq}   \text{beq} \} [\text{label}] \text{ term} = \text{term} ":"$	
<i>cequation</i>	$::= \{ \text{ceq}   \text{bceq} \} [\text{label}] \text{ term} = \text{term if term} ":"$	
<i>transition</i>	$::= \{ \text{trans}   \text{btrans} \} [\text{label}] \text{ term} \Rightarrow \text{term} ":"$	
<i>ctransition</i>	$::= \{ \text{ctrans}   \text{bctrans} \} [\text{label}] \text{ term} \Rightarrow \text{term if term} ":"$	
<i>fol</i>	$::= \text{ax} [\text{label}] \text{ term} ":"$	
<i>label</i>	$::= "[" \text{ ident} \dots "]:"$	
<i>macro</i>	$::= \#define \text{ term} ::=\text{ term} ":"$	

<sup>1</sup>The nonterminal *ident* is for identifiers and will be defined in the section 2.3.

<sup>2</sup>*module\_expr* is defined in the section 1.3.

<sup>3</sup>If optional *[ protecting | extending | including ]* is omitted, it is defaulted to *protecting*.

<sup>4</sup>*comment* is discussed in section 2.5.

<sup>5</sup>There must not be any separators (see section 2) between *ident* and *qualifier*.

<sup>6</sup>*operator\_symbol* is defined in section 2.4.

<sup>7</sup>*natural* is a natural number written in ordinal arabic notation.

## 1.3 Module Expression

```

module_expr ::= module_name | sum | rename | instantiation | "("module_expr")"
sum          ::= module_expr { + module_expr } ...
rename       ::= module_expr * {"rename_map, ..."}
instantiation ::= module_expr ("{" ident[qualifier] <= aview}, ... ")
rename_map  ::= sort_map | op_map
sort_map    ::= { sort | hsort } sort_name -> ident
op_map      ::= { op | bop } op_name -> operator_symbol
op_name     ::= operator_symbol | "("operator_symbol")" qualififier
aview        ::= view_name | module_expr
               | view to module_expr {"view_elt, ... "}
view_name   ::= ident
view_elt    ::= sort_map | op_view | variable
op_view     ::= op_map | term -> term

```

When a module expression is not fully parenthesized, the proper nesting of subexpressions may be ambiguous. The following precedence rule is used to resolve such ambiguity:

*sum < rename < instantiation*

## 1.4 View Declaration

```

view ::= view view_name from module_expr to module_expr
      {"view_elt, ... "}

```

## 1.5 Evaluation

```

eval    ::= { reduce | behavioural-reduce | execute } context term "."
context ::= in module_expr :

```

The interpreter has a notion of *current module* which is specified by a *module\_expr* and establishes a context. If it is set, *context* can be omitted.

## 1.6 Sugars and Abbreviations

**Module type** There are following abbreviations for *module\_type*.

Keyword	Abbriviation
module	mod
module!	mod!
module*	mod*

### Module Declaration

*make* ::= make *module\_name* "(" *module\_expr* ")"

*make* is a short hand for declaring module of name *module\_name* which imports *module\_expr* with protecting mode.

make FOO (BAR \* {sort Bar -> Foo})

is equivalent to

module FOO { protecting (BAR \* {sort Bar -> Foo}) }

**Principal Sort** `principal-sort` can be abbreviated to `psort`.

**Import Mode** For import modes, the following abbreviations can be used:

Keyword	Abbriviation
protecting	pr
extending	ex
including	inc
using	us

**Simultaneous Operator Declaration** Several operators with the same arity, coarity and operator attributes can be declared at once by `ops`. The form

`ops operator_symbol1 ... operator_symboln : arity -> coarity op_attrs`

is just equivalent to the following multiple operator declarations:

`op operator_symbol1 : arity -> coarity op_attrs`

$\vdots$

`op operator_symboln : arity -> coarity op_attrs`

`bops` is the counterpart of `ops` for behavioural operators.

`bops operator_symbol ... : arity -> coarity op_attrs`

In simultaneous declarations, parentheses are sometimes necessary to separate operator symbols. This is always required if an operator symbol contains dots, blank characters or underscores.

**Predicate** Predicate declaration (*predicate*) is a syntactic sugar for declaring `Bool` valued operators, and has the syntax:

`predicate ::= pred operator_symbol : arity [ op_attrs ]` –<sup>8</sup>

The form

`pred operator_symbol : arity op_attrs`

is equivalent to:

`op operator_symbol : arity -> Bool op_attrs`

**Operator Attributes** The following abbreviations are available:

Keyword	Abbriviation
associative	assoc
commutative	comm
idempotent	idem

<sup>8</sup>You cannot use `sort_name` of the same character sequence as that of any keywords, i.e., `module`, `op`, `vars`, etc. in `arity`.

**Axioms** For the keywords introducing axioms, the following abbreviations can be used:

Keyword	Abbriviation	Keyword	Abbriviation
ceq	cq	bceq	bcq
trans	trns	ctrans	ctrns
btrans	btrns	bctrans	bctrns

**Blocks of Declarations** References to (importations of) other modules, signature definitions and axioms can be clustered in blocked declarations:

```

imports ::= imports "{"
           { import | comment } ...
           "}"
signature ::= signature "{"
             { sort | record | operator | comment } ...
             "}"
axioms   ::= axioms "{"
             { variable | axiom | comment } ...
             "}"

```

**Views** To reduce the complexity of views appearing in module instantiation, some sugars are provided.

First, it is possible to identify parameters by positions, not by names. For example, if a parameterized module is declared like

```
module! FOO (A1 :: TH1, A2 :: TH2) { ... }
```

the form

```
FOO(V1, V2)
```

is equivalent to

```
FOO(A1 <= V1, A2 <= V2)
```

Secondly, view to construct in arguments of module instantiations can always be omitted. That is,

```
FOO(A1 <= view to module_expr{...})
```

can be written as

```
FOO(A1 <= module_expr{...})
```

## Evaluation

Keyword	Abbriviation
reduce	red
bereduce	bred
execute	exec

## 2 Lexical Considerations

A **CafeOBJ** spec is written as a sequence of tokens and separators. A *token* is a sequence of “printing” ASCII characters (octal 40 through 176).<sup>9</sup> A *separator* is a “blank” character (space, vertical tab, horizontal tab, carriage return, newline, form feed). In general, any number of separators may appear between tokens.

---

<sup>9</sup>The current interpreter accepts Unicode characters also, but this is beyond the definition of CafeOBJ language.

## 2.1 Reserved Word

There are *no* reserved word in **CafeOBJ**. One can use keywords such as `module`, `op`, `var`, or `signature`, etc. for identifiers or operator symbols.

## 2.2 Self-terminating Characters

The following eight characters are always treated as *self-terminating*, i.e., the character itself construct a token.

( ) , [ ] { } ;

## 2.3 Identifier

Nonterminal *ident* is for *identifier* which is a sequence of any printing ASCII characters except the followings:

self-terminating characters (see section 2.2)  
. (dot)  
" (double quote)

Upper- and lowercase are distinguished in identifiers. *idents* are used for module names (*module\_name*), view names (*view\_name*), parameter names (*parameter\_name*), sort symbols (*sort\_symbol*), variables(*var\_name*), slot names (*slot\_name*) and labels (*label*).

## 2.4 Operator Symbol

The nonterminal *operator\_symbol* is used for naming operators (*operator*) and is a sequence of any ASCII characters (self-terminating characters or non-printing characters can be an element of operator names.)<sup>10</sup>

Underscores are specially treated when they appear as a part of operator names; they reserve the places where arguments of the operator are inserted. Thus the single underscore cannot be a name of an operator.

## 2.5 Comments and Separators

A *comment* is a sequence of characters that begins with one of the following four character sequences

-- -->  
\*\* \*\*>

which ends with a newline character, and contains only printing ASCII characters and horizontal tabs in between.

A *separator* is a blank character (space, vertical tab, horizontal tab, carriage return, newline, from feed). One or more separators must appear between any two adjacent non-self-terminating tokens.<sup>11</sup>

Comments also act as separators, but their appearance is limited to some specific places (see section 1).

---

<sup>10</sup>The current implementation does not allow EOT character (control-D) to be an element of operator symbol.

<sup>11</sup>The same rule is applied to *term*. Further, if an *operator\_symbol* contains blanks or self-terminating characters, it is sometimes necessary to enclose a term with such operator as top by parentheses for disambiguation.